
標準部品化力、オブジェクトおよび状態(第 3 章) 

3.3 可変データのモデル化 

3.3.2 キューの表現 

set-car!やset-cdr!を用いることで、キューを作成することができるようになります。

次々とやってくるデータを次々と追加し、必要に応じて追加したデータを取り出せるリ

ストがあると便利ですが、追加するときは末尾に追加し、取り出すときは必ず先頭から

取り出すようにしたリストのことをキューと呼びます。キューは、最初に入れたものが

最初にでてくるので、FIFO (first in first out)とも呼ばれます。 

 

キューに効率を良くするために、先頭と、末尾のデータを指すポインタを持つように

します。 

まず、基本的な関数です。 

(define (make-queue) (cons '() '())) 

(define (front-ptr queue) (car queue)) 

(define (rear-ptr queue) (cdr queue)) 

(define (set-front-ptr! queue item) (set-car! queue item)) 

(define (set-rear-ptr! queue item) (set-cdr! queue item)) 

このように set-car!や set-cdr!を使うと、関数を返すことによってオブジェクトを作

成することに比べかなり容易にデータオブジェクトがつくれる、ということがわかりま

す。 

 続いて、空であるかどうかを判定するためには、front-ptrが空であるかどうかを判

定すれば良いので、 

(define (empty-queue? queue) (null? (front-ptr queue))) 

となって、先頭の要素を見るためには、front-ptr が指すリストの先頭の要素を返せば

良い、ということになります。 

(define (front-queue queue) 

  (if (empty-queue? queue) 

      (error "FRONT called with an empty queue" queue) 

      (car (front-ptr queue)))) 

次に、挿入ですが、キューは常に末尾にデータが加わるので、set-rear-ptr!で末尾に

新しい要素を追加してやれば良い、ということになります。例えば、p.1の図に

(insert-queue! q 'd)を実行すると次のような状態になります。 

 

また、キューが空である場合には特殊な操作が必要で、front-ptrと rear-ptr が同じ

新しいアイテムを指すようにする必要があります。 

(define (insert-queue! queue item) 

  (let ((new-pair (cons item '()))) 

    (cond ((empty-queue? queue) 

           (set-front-ptr! queue new-pair) 

           (set-rear-ptr! queue new-pair) 

           queue) 

          (else 

           (set-cdr! (rear-ptr queue) new-pair) 

           (set-rear-ptr! queue new-pair) 

           queue)))) 

 最後に deleteは、先頭の要素を削除すれば良いので、set-front-ptr!で、front-ptr

を一つ後ろにつなげなおせば okです。例えば、p.2の図のキューに(delete-queue! q)

を実行すると次のような状態になります。 

(define (delete-queue! queue) 

  (cond ((empty-queue? queue) 

         (error "DELETE! called with an empty queue" queue)) 

        (else 

         (set-front-ptr! queue (cdr (front-ptr queue))) 

         queue))) 

 

3.3.3 表の表現 

次に連想リスト(キーと値による 1次元の表)のオブジェクトをつくることを考えてみ

ます。連想リストが空になったときに困らないように、ダミーのエントリーもつくって

おきます。 

 

ここでは、*table*とあるのがダミーのエントリーで、table全体はリストになってい

ます。先頭の要素がダミーで、次の要素から、キーと値のペアが格納されている、とい

うことになります。このようなリストは頭つきリスト(headed list)と呼ばれ、*table*

をもつペアが頭(head)で、それに続くリストは背骨(backbone)と呼ばれます。上の table

は a: 1, b: 2, c: 3という連想リストを表現しています。 

 ダミーは連想リストが空になったときや、リストの先頭の要素を削除したときに困ら

ないようにするために用意されていますが、ある意味、ダミーのエントリーが連想リス

トを表すデータ構造の本体、つまり連想リストオブジェクトと考えることもできます。

このようなダミーがないといろいろと不便が生じます。例えば、仮に table が上の

((a.1) (b.2) (c.3))を指していたとします。ここで、(a.1)を削除しようとしたら、ど

うなるでしょう？プログラムを書いてみるとわかりやすいのですが、tableという変数

を set!で書き換えないといけないのですが、連想リストを指す変数名は tableだけと

は限らないので、単に(set! table (cdr table))とすればいいわけではありません。つ

まり、どんな名前の連想リストに対してもその連想リストを操作できる関数を記述しな

ければいけないのですが、それは非常に難しいです(Schemeではおそらくできない)。

上記のダミーがこのような問題を解決するのに大変有効と言えます。 

このようなデータに対する検索は次の lookupで実現できます。 

(define (myassoc key records) 

  (cond ((null? records) #f) 

        ((equal? key (car (car records))) (car records)) 

        (else (myassoc key (cdr records))))) 

(define (lookup key table) 

  (let ((record (myassoc key (cdr table)))) 

    (if record 

        (cdr record) 

        #f))) 

等しいキーがあるかどうか順に探していきます。 

新しいエントリーを挿入するプログラムは次のようになります。 

(define (insert! key value table) 

  (let ((record (myassoc key (cdr table)))) 

    (if record 

        (set-cdr! record value) 

        (set-cdr! table 

                  (cons (cons key value) (cdr table))))) 

  'ok) 

最後に、新しい連想リストを作るプログラムは次のようになります。 

(define (make-table) (list '*table*)) 

実行例は次のようになります。 

> (define a (make-table)) 

> a 

(*table*) 

> (insert! 'a 1 a) 

ok 

> a 

(*table* (a . 1)) 

> (insert! 'b 2 a) 

ok 

> (insert! 'c 3 a) 

ok 

> (lookup 'b a) 

2 

> a 

(*table* (c . 3) (b . 2) (a . 1)) 

 

クォートによるリストと LIST 手続きの違い 

 クォートによるリストと list関数によって得られるリストの違いを認識していない

と問題になることがあります。 

クォートによって得られるリスト:  '(a b c) 

list 関数によって得られるリスト: (list 'a 'b 'c) 

どちらも同じものを返すようにみえますが、実は大きな違いがあります。クォートによ

って作られるリストはプログラム上にただ１カ所にしか存在せず、そのポインタが返さ

れます。一方 list関数は、そこに新しいリスト構造を生成する、ということを行いま

す。例えば、次のプログラムと実行例をみてください。 

(define (f) '(a b c)) 

(define (g) (list 'a 'b 'c)) 

(define x (list (f) (f) (f))) 

(define y (list (g) (g) (g))) 

> x 

((a b c) (a b c) (a b c)) 

> y 

((a b c) (a b c) (a b c)) 

> (set-car! (car x) 'change) 

> (set-car! (car y) 'change) 

> x 

((change b c) (change b c) (change b c)) 

> y 

((change b c) (a b c) (a b c)) 

クォートで作成した xのほうは全ての'aが'changeに置き換わったのに、list関数で

作成した yのほうは先頭の'aだけが'changeに置き換わっています。 

 これは、上記のクォートで生成された xは次のようなデータ構造になっていて、 

 

list 手続きによって生成された yは次のようなデータ構造になっているためです。 

 

さらにここで fを実行してみます。 

> (f) 

(change b c) 

fという関数の中のリストも変化していることがわかります。 

この現象を理解するのはちょっと難しいですが、次のように考えれば良いと思います。

まず、プログラムをインタプリタが読み込みます。プログラムを読み込むとき(define 

(f x)…)といった関数の定義を読みこむと、そのときに関数を表す関数オブジェクトが

生成されます。その関数の中にクォートによるリストが記述されてあった場合はクォー

トによるリストは関数と一緒に生成されて、関数の中に埋め込まれます。よって、なん

らかの関数が返すクォートによるリストは同一のリストが返されるようになります。一

方、list関数によるリストは、プログラム読み込み時には、まだリストが生成されて

いません。インタプリタ上で、このプログラムを実行させ、その list 関数が実行され

x

a b c

y

a b c a b c a b c

た時に初めてリストオブジェクトが生成されます。また、このとき同一のリストオブジ

ェクトを返さず、新しいリストオブジェクトを動的に生成します。 

一言で言うと、クォートによるリストは静的に生成されるリストで、list関数によ

るリストは動的に生成されるリストということもできます。 

